

# NAG Library Routine Document

## G08AEF

**Note:** before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

### 1 Purpose

G08AEF performs the Friedman two-way analysis of variance by ranks on  $k$  related samples of size  $n$ .

### 2 Specification

```
SUBROUTINE G08AEF (X, LDX, K, N, W1, W2, FR, P, IFAIL)
```

```
INTEGER LDX, K, N, IFAIL
```

```
REAL (KIND=nag_wp) X(LDX,N), W1(K), W2(K), FR, P
```

### 3 Description

The Friedman test investigates the score differences between  $k$  matched samples of size  $n$ , the scores in the  $i$ th sample being denoted by

$$x_{i1}, x_{i2}, \dots, x_{in}.$$

(Thus the sample scores may be regarded as a two-way table with  $k$  rows and  $n$  columns.) The hypothesis under test,  $H_0$ , often called the null hypothesis, is that the samples come from the same population, and this is to be tested against the alternative hypothesis  $H_1$  that they come from different populations.

The test is based on the observed distribution of score rankings between the matched observations in different samples.

The test proceeds as follows

(a) The scores in each column are ranked,  $r_{ij}$  denoting the rank within column  $j$  of the observation in row  $i$ . Average ranks are assigned to tied scores.

(b) The ranks are summed over each row to give rank sums  $t_i = \sum_{j=1}^n r_{ij}$ , for  $i = 1, 2, \dots, k$ .

(c) The Friedman test statistic  $F$  is computed, where

$$F = \frac{12}{nk(k+1)} \sum_{i=1}^k \left\{ t_i - \frac{1}{2}n(k+1) \right\}^2.$$

G08AEF returns the value of  $F$ , and also an approximation,  $p$ , to the significance of this value. ( $F$  approximately follows a  $\chi_{k-1}^2$  distribution, so large values of  $F$  imply rejection of  $H_0$ ).  $H_0$  is rejected by a test of chosen size  $\alpha$  if  $p < \alpha$ . The approximation  $p$  is acceptable unless  $k = 4$  and  $n < 5$ , or  $k = 3$  and  $n < 10$ , or  $k = 2$  and  $n < 20$ ; for  $k = 3$  or  $4$ , tables should be consulted (e.g., Siegel (1956)); for  $k = 2$  the Sign test (see G08AAF) or Wilcoxon test (see G08AGF) is in any case more appropriate.

### 4 References

Siegel S (1956) *Non-parametric Statistics for the Behavioral Sciences* McGraw-Hill

## 5 Parameters

- 1: X(LDX,N) – REAL (KIND=nag\_wp) array *Input*  
*On entry:* X( $i, j$ ) must be set to the value,  $x_{ij}$ , of observation  $j$  in sample  $i$ , for  $i = 1, 2, \dots, k$  and  $j = 1, 2, \dots, n$ .
- 2: LDX – INTEGER *Input*  
*On entry:* the first dimension of the array X as declared in the (sub)program from which G08AEF is called.  
*Constraint:* LDX  $\geq$  K.
- 3: K – INTEGER *Input*  
*On entry:*  $k$ , the number of samples.  
*Constraint:* K  $\geq$  2.
- 4: N – INTEGER *Input*  
*On entry:*  $n$ , the size of each sample.  
*Constraint:* N  $\geq$  1.
- 5: W1(K) – REAL (KIND=nag\_wp) array *Workspace*  
 6: W2(K) – REAL (KIND=nag\_wp) array *Workspace*
- 7: FR – REAL (KIND=nag\_wp) *Output*  
*On exit:* the value of the Friedman test statistic,  $F$ .
- 8: P – REAL (KIND=nag\_wp) *Output*  
*On exit:* the approximate significance,  $p$ , of the Friedman test statistic.
- 9: IFAIL – INTEGER *Input/Output*  
*On entry:* IFAIL must be set to 0,  $-1$  or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.  
 For environments where it might be inappropriate to halt program execution when an error is detected, the value  $-1$  or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value  $-1$  or 1 is used it is essential to test the value of IFAIL on exit.**  
*On exit:* IFAIL = 0 unless the routine detects an error or a warning has been flagged (see Section 6).

## 6 Error Indicators and Warnings

If on entry IFAIL = 0 or  $-1$ , explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL = 1

On entry, N < 1.

IFAIL = 2

On entry, LDX < K.

IFAIL = 3

On entry,  $K \leq 1$ .

## 7 Accuracy

For estimates of the accuracy of the significance  $p$ , see G01ECF. The  $\chi^2$  approximation is acceptable unless  $k = 4$  and  $n < 5$ , or  $k = 3$  and  $n < 10$ , or  $k = 2$  and  $n < 20$ .

## 8 Further Comments

The time taken by G08AEF is approximately proportional to the product  $nk$ .

If  $k = 2$ , the Sign test (see G08AAF) or Wilcoxon test (see G08AGF) is more appropriate.

## 9 Example

This example is taken from page 169 of Siegel (1956). The data relates to training scores of three matched samples of 18 rats, trained under three different patterns of reinforcement.

### 9.1 Program Text

```

Program g08aefe

!      G08AEF Example Program Text

!      Mark 24 Release. NAG Copyright 2012.

!      .. Use Statements ..
      Use nag_library, Only: g08aef, nag_wp, x04caf
!      .. Implicit None Statement ..
      Implicit None
!      .. Parameters ..
      Integer, Parameter          :: nin = 5, nout = 6
!      .. Local Scalars ..
      Real (Kind=nag_wp)          :: fr, p
      Integer                     :: i, ifail, k, ldx, n
!      .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable :: w1(:), w2(:), x(:, :)
!      .. Executable Statements ..
      Write (nout,*) 'G08AEF Example Program Results'
      Write (nout,*)

!      Skip heading in data file
      Read (nin,*)

!      Read in the problem size
      Read (nin,*) k, n

      ldx = k
      Allocate (x(ldx,n),w1(k),w2(k))

!      Read in data
      Read (nin,*)(x(i,1:n),i=1,k)

!      Display title
      Write (nout,*) 'Friedman test'
      Write (nout,*)
      Flush (nout)

!      Display input data
      ifail = 0
      Call x04caf('General',' ',k,n,x,ldx,'Data values',ifail)

!      Perform ANOVA
      ifail = 0

```

```

      Call g08aef(x,ldx,k,n,w1,w2,fr,p,ifail)

!      Display results
      Write (nout,*)
      Write (nout,99999) 'Test statistic          ', fr
      Write (nout,99998) 'Degrees of freedom     ', k - 1
      Write (nout,99999) 'Significance          ', p

99999 Format (1X,A,F6.3)
99998 Format (1X,A,I6)
      End Program g08aefe

```

## 9.2 Program Data

```

G08AEF Example Program Data
3  18                                :: K,N
1.0 2.0 1.0 1.0 3.0 2.0 3.0 1.0 3.0
3.0 2.0 2.0 3.0 2.0 2.5 3.0 3.0 2.0
3.0 3.0 3.0 2.0 1.0 3.0 2.0 3.0 1.0
1.0 3.0 3.0 2.0 3.0 2.5 2.0 2.0 3.0
2.0 1.0 2.0 3.0 2.0 1.0 1.0 2.0 2.0
2.0 1.0 1.0 1.0 1.0 1.0 1.0 1.0 1.0  :: End of X

```

## 9.3 Program Results

G08AEF Example Program Results

Friedman test

Data values

	1	2	3	4	5	6	7
1	1.0000	2.0000	1.0000	1.0000	3.0000	2.0000	3.0000
2	3.0000	3.0000	3.0000	2.0000	1.0000	3.0000	2.0000
3	2.0000	1.0000	2.0000	3.0000	2.0000	1.0000	1.0000
	8	9	10	11	12	13	14
1	1.0000	3.0000	3.0000	2.0000	2.0000	3.0000	2.0000
2	3.0000	1.0000	1.0000	3.0000	3.0000	2.0000	3.0000
3	2.0000	2.0000	2.0000	1.0000	1.0000	1.0000	1.0000
	15	16	17	18			
1	2.5000	3.0000	3.0000	2.0000			
2	2.5000	2.0000	2.0000	3.0000			
3	1.0000	1.0000	1.0000	1.0000			

```

Test statistic      8.583
Degrees of freedom    2
Significance        0.014

```

---